

# Chapter 1

## Maple Fundamentals

The purpose of this tutorial is to introduce some basic Maple commands, syntax, and programming concepts for Maple V 9.5. This tutorial is intended for the beginning calculus student. It is by no means thought of as a comprehensive introduction to Maple. We will illustrate how to get started with Maple and introduce some of the important commands which are very useful in calculus. Maple was developed at the University of Waterloo in Ontario, Canada, which explains the Maple leaf.

To begin, we introduce the **restart** command, which we enter to ensure all memory is cleared.

```
> restart;
```

### 1.1 Basics

After you log on to your terminal and access Maple, you will get a window containing a "prompt" `>` where you can immediately begin entering Maple commands. If you wanted to multiply the numbers 247 and 3756, you would enter

```
> 247*3756;
```

927732

Like every programming language there are a few rules which you have to follow. These are a few of them:

1. Maple is a computer algebra system. It will attempt to give the answers in exact symbolic form, rather than finding numerical approximations.
2. Every Maple V command must end with either a semicolon or a colon. If a command is ended with a colon then the calculation will be made but no output will be printed. When a semicolon is used, the calculation will be made and the result will be printed.
3. Use the `#` sign for comments.
4. Maple is case sensitive. It distinguishes between upper and lower case letters. For example, `X` and `x` mean different things.
5. Operators for arithmetic are required for all expressions. For example, a `*` for multiplication is required. This is different from graphing calculators where you can enter `2x`. In Maple `2*x` is required.
6. Use `%` to refer to the last executed expression.
7. The `?` command can be used to obtain help on most commands.
8. The major difference between `=` and `:=`. `=` sign is used in equations, the two expressions on each side have to balance. The sequence `:=` is used to define new quantities, almost like an abbreviation or a shortcut.
9. Defining the constants  $\pi$  and  $e$

We illustrate each of these important facts separately.

### 1.1.1 Maple is a Computer Algebra System and Will Express Answers in Exact Symbolic Form

Like other computer algebra systems, Maple uses exact arithmetic. For example, if you divide two integers, Maple V will return the exact answer.

```
> 3235/7478;
```

$$\frac{3235}{7478}$$

To get the decimal representation, one can simply end one of the numbers with a decimal, such as

```
> 3235./7478;
```

$$0.4326023001$$

Another way to get a decimal number is to enclose the calculation in the **evalf** command.

```
> evalf(3235/7478);
```

$$0.4326023001$$

The following two commands obtain the exact and numerical approximation of the following fraction addition:

```
> 25/27 + 3/51;
```

$$\frac{452}{459}$$

```
> evalf(25/27 + 3/51);
```

$$0.9847494553$$

Compare the following ways using the **sqrt** command to find the  $\sqrt{8}$ .

```
> sqrt(8); sqrt(8.); evalf(sqrt(8));
```

$$2\sqrt{2}$$
$$2.828427125$$
$$2.828427124$$

### 1.1.2 Every Maple V Command Must End With Either a Semicolon or a Colon

This is a fundamental property concerning all Maple commands. The following command adds 247 and 3756

```
> 247 + 3756;
```

$$4003$$

If a colon were used on the previous command, the result would look like the following (note there is no output):

```
> 247 + 3756:
```

A nice feature of Maple Version 9.5 is that if you forget to end a command with a semi-colon or colon, Maple will add a semi-colon for you as the next command illustrates. This feature was not available in earlier versions.

```
> 247 + 3756
```

```
Warning, inserted missing semicolon at end of statement, 247 +  
3756;
```

$$4003$$

However, because Maple allows full screen editing, you can go back to the line where the problem occurred and correct it.

### 1.1.3 Use the # Sign for Comments

Comments in Maple commands can be useful for describing what a command does. Comments are added with the # sign. Anything typed in to right of the # symbol will be treated as a comment and will not be executed. The following commands illustrates this fact.

```
> 247*3756; # Multiply 247 and 3756
```

$$927732$$

```
> 247-3756; # Subtract 3756 from 247
```

$$-3509$$

### 1.1.4 Maple is Case Sensitive

A common problem people often have when first using Maple is realizing that Maple is case sensitive. Lower and upper case letters are treated as entirely different entities. To illustrate this, see if you can understand why the next two commands produced entirely different answers.

```
> x-x; # Subtracting two lower case x's
0
> X-x; # Subtracting an upper and lower case X.
X - x
```

This fact carries over to all Maple commands. Later in this tutorial, we will describe how to factor polynomials using the **factor** command. See if you can compare the next two commands and understand why one produces the correct answer and the other does not when factoring  $x^2 - 4$ .

```
> factor(x^2-4);
(x - 2)(x + 2)
> FACTOR(x^2-4);
FACTOR(x^2 - 4)
```

Even having one letter with the wrong case will not produce the desired result

```
> Factor(x^2-4);
Factor(x^2 - 4)
```

### 1.1.5 Operators for Arithmetic are Required for all Expressions. For Example, a \* for Multiplication is Required

A feature that often confuses students when first learning Maple syntax is that operators are required when executing arithmetic expressions. Operators are not always required when using graphics calculators or WeBWorK. However, they must be included when using Maple to achieve the desired results.

The basic arithmetic operations of addition, multiplication, division, and exponentiation are recognized in Maple V by the following symbols:

+	and -	add and subtract
*	and /	multiply and divide
^	or **	raise to a power

Often, the operator ignored by students is the multiplication \* operator. For example, the following command multiplies 3 and 5.

```
> 3*5;
15
```

However, without the \* operator, Maple does not know multiplication is implied as the next command illustrates

```
> (3)(5);
3
```

We will learn below that the := is used to assign quantities to variables. Suppose we assign the variable x the value of 5

```
> x := 5;
x := 5
```

You should be able to compare the following results and understand why only the first gives a result of 10 and why the second generates an error.

```
> 2*x;
10
> 2x;
Error, missing operator or ';'
3
```

To unassign  $x$ , we assign it to itself enclosed in single quotes.

```
> x := 'x';  
x := x
```

Thus,  $x$  longer has a value of 5.

```
> 2*x;  
2x
```

You can raise a number to a power as follows:

```
> 2^6;  
64  
> 2**6;  
64
```

Using **restart**, we can unassign all previously defined variables.

```
> restart;
```

### 1.1.6 Use % to Refer to the Last Executed Expression.

Operations can be performed on previous results by using the percent symbol  $\%$ . To illustrate how this works, suppose we enter the following:

```
> 3+5;  
8
```

To multiply the previous result by 10 without retyping the 8, we can enter

```
> % * 10;  
80
```

The previous result is now 80. We can add 20 to this by entering

```
> 20 + %;  
100
```

Previous results from more than one command before can be accessed by typing more than one  $\%$  sign. For example, to multiply 80 that was output two results back by 2, we could enter

```
> 2 * %%;  
160
```

### 1.1.7 The ? Command can be Used to Obtain Help on Most Commands

You can obtain the help file regarding most Maple functions and statements by entering  $? \text{ name}$  where  $\text{name}$  is the function or statement you desire help on. For example, to see help on **sqrt**, you can enter:

```
> ?sqrt
```

### 1.1.8 The Major Difference Between = and :=

The  $=$  sign is used in equations, the two expressions on each side have to balance. The sequence  $:=$  is used to assign quantities to variables, almost like an abbreviation or a shortcut. For example, here is an equation.

```
> x^2 - 4*x + 3 = 0;  
x^2 - 4x + 3 = 0
```

We can also do this

```
> y := x^2 - 4*x + 3;  
y := x^2 - 4x + 3
```

Here we set  $y$  equal to the expression  $x^2 - 4x + 3$ . From now on  $y$  will stand for this expression. For example  $y^2$  yields

```
> y^2;  
(x^2 - 4x + 3)^2
```

and the first equation can be expressed as

```
> y = 0;
```

$$x^2 - 4x + 3 = 0$$

Suppose now we assign the variable x to a number.

```
> x := 5;
```

$$x := 5$$

If you forget what is stored in a variable, you can simply enter the variable you want.

```
> x;
```

$$5$$

This would be the value stored in y.

```
> y;
```

$$8$$

Why is y not the original quantity that was assigned? The reason is for  $y := x^2 - 4x + 3$  entered above, x has been replaced with 5 and the expression has been numerically evaluated. If this is not desired, it can be fixed by unassigning the variable. Assigning a variable to itself enclosed in single quotes unassigns the variable. Thus, to unassign x, enter

```
> x := 'x';
```

$$x := x$$

The next command shows that both the assigned values for x and y have been updated

```
> x; y;
```

$$\begin{array}{c} x \\ x^2 - 4x + 3 \end{array}$$

In order to undo all definitions issue the **restart** command.

```
> restart;
```

```
> x; y;
```

$$x$$

$$y$$

### 1.1.9 Defining the Constants $\pi$ and $e$

In this class, it will be important to use the numbers  $\pi$  and  $e$ . The number  $\pi$  is entered as Pi (note the upper case P). To enter  $\pi$ , see its decimal approximation, and see it evaluated with the cosine function, we enter

```
> Pi; evalf(Pi); cos(Pi);
```

$$\begin{array}{c} \pi \\ 3.141592654 \\ -1 \end{array}$$

Entering pi with a lower case will only express the symbol for  $\pi$  - it DOES NOT represent the number.

```
> pi; evalf(pi); cos(pi);
```

$$\begin{array}{c} \pi \\ \pi \\ \cos(\pi) \end{array}$$

To enter the number e, we use the **exp** command. Its argument requires entering the exponent the number e is raised to enclosed in parenthesis. The following command demonstrates how to enter this number correctly.

```
> exp(1); evalf(exp(1));
```

$$\begin{array}{c} e \\ 2.718281828 \end{array}$$

The next commands illustrate common ways this number is not entered correctly.

```
> e^1; evalf(e^1); exp^1); # These ways are not correct
```

$$\begin{array}{c} e \\ e \\ \exp \end{array}$$

$$5$$

## 1.2 Simple Algebra

Here are some of the important Maple V algebra commands:

<u>Command</u>	<u>What it does</u>
<b>expand</b>	Multiples and expands algebraic expressions
<b>simplify</b>	Simplifies algebraic expressions
<b>factor</b>	solves equations, displays exact solutions
<b>solve</b>	solves equations, displays exact solutions
<b>fsolve</b>	solves equations, displays numerical solutions

Some examples of these commands follow:

To factor the polynomial  $x^2 - 4x + 3$ , we enter

```
> factor(x^2 -4*x + 3);
```

$$(x - 1)(x - 3)$$

The next two commands store and expand the expression  $(x - 1)(x - 2)(x - 3)(x - 4)$

```
> f := (x-1)*(x-2)*(x-3)*(x-4);
```

$$f := (x - 1)(x - 2)(x - 3)(x - 4)$$

```
> expand(f); # maple uses the last definition of f
```

$$x^4 - 10x^3 + 35x^2 - 50x + 24$$

Suppose we enter the following expression:

```
> (x^(-1) + y^(-1))^(-1);
```

$$\frac{1}{\frac{1}{x} + \frac{1}{y}}$$

The next command uses the **simplify** command and simplifies the previous result.

```
> simplify(%);
```

$$\frac{xy}{y+x}$$

A great deal of algebra deals with solving equations. In Maple we have the **solve** command to accomplish this task. The syntax is **solve**(*equation*, *variable*), that is, you need to communicate the equation and the variable which you try to find. The **evalf** command frequently comes in handy with expressing numerical values in equations.

As a first example, suppose we want solve the equation  $x^3 + x + 1 = 3x^2$  for  $x$ . Note the equation and how it is stored in the variable `e1`.

```
> e1 := x^3+x+1 = 3*x^2; # define e1 to be the equation in question
```

$$e1 := x^3 + x + 1 = 3x^2$$

```
> solve(e1 ,x); evalf(%);
```

$$1, 1 + \sqrt{2}, 1 - \sqrt{2}$$
$$1., 2.414213562, -0.414213562$$

We can also use solve equations with multiple variables. To demonstrate, suppose we want to solve for  $x$  for the equation  $y = \frac{x-1}{x-2}$ . The next two commands enter this equation and solve for the indicated variable:

```
> y = (x-1)/(x-2);
```

$$y = \frac{x-1}{x-2}$$

```
> solve(%,x);
```

$$\frac{2y-1}{y-1}$$

We next use **solve** to solve the equation  $\cos(x) = \sin(2x)$  for  $x$ .

```
> solve(cos(x)=sin(2*x),x);
```

$$-\frac{1}{2}\pi, \frac{1}{2}\pi, \frac{1}{6}\pi, \frac{5}{6}\pi$$

We can obtain the decimal representations as follows

```
> evalf(%);
```

-1.570796327, 1.570796327, 0.5235987758, 2.617993878

Note that the last command only finds four of many solutions to the equation.

A variation of the solve command is **fsolve**. It provides numerical solutions directly

```
> fsolve (cos(x)=sin(2*x),x);
```

1.570796327

Note that we only found one solution. A search interval for the solution can be added as an option. To search for a solution in the interval  $[\pi, 2\pi]$ , enter

```
> fsolve( cos(x)=sin(2*x),x, Pi..2*Pi);
```

4.712388980

## 1.3 Functions

We now demonstrate how Maple can be used to define functions. First, we make sure all variables are unassigned:

```
> restart;
```

There are two ways to define and work with functions. One way is to define the function as an expression. Consider the function  $f(x) = x^2$ . We enter this as an expression.

```
> f := x^2;
```

$f := x^2$

This definition can be checked by entering

```
> f;
```

$x^2$

The Maple V command **subs** allows expressions such as this to be evaluated at specific values. The next command uses **subs** to substitute  $x = 5$  into the given function.

```
> subs(x=5,f);
```

25

The problem with defining functions as expressions is that standard functional notation, such as  $f(x)$  or  $f(5)$ , is not understood by Maple and results in nonsense.

```
> f(x);
```

$x(x)^2$

```
> f(5);
```

$x(5)^2$

If you wish to use standard functional notation, you must enter the function using the minus-greater than notation  $\rightarrow$ , made by typing the "minus sign" followed by the "greater than" sign. For example:

```
> f := x -> x^2;
```

$f := x \rightarrow x^2$

Once this is done, we can evaluate functions in the normal way.

```
> f(x);
```

$x(x)^2$

```
> f(5);
```

25

It is essential that you understand the difference between functions, expressions and equations. Let's do this by example.

```
> x^2 - 4*x + 3; # this is an expression
```

```
> y = x^2 - 4*x + 3; # this is an equation, y equals something
```

```
> y := x^2 - 4*x + 3; # define y to be the expression x^2 - 4*x + 3
```

```
> f := x -> x^2 - 4*x + 3; # define f to be the function with input x and output x^2 - ...
```

$x^2 - 4x + 3$

$y = x^2 - 4x + 3$

$y := x^2 - 4x + 3$

$f := x \rightarrow x^2 - 4x + 3$

Notice that  $y$  and  $f(x)$  are the same, but not  $y$  and  $f$  !!!!!

```
> y; f(x); f;
```

$$x^2 - 4x + 3$$

$$x^2 - 4x + 3$$

$$f$$

The function notation allows us to evaluate many things, as the following examples illustrate.

```
> f(x);
> f(a);
> f(-x);
> f(x-3); simplify(%);
> (f(x+h)-f(x))/h; simplify(%);
```

$$x^2 - 4x + 3$$

$$a^2 - 4a + 3$$

$$x^2 + 4x + 3$$

$$(x - 3)^2 - 4x + 15$$

$$x^2 - 10x + 24$$

$$\frac{(x + h)^2 - 4h - x^2}{h}$$

$$2x + h - 4$$

These functions are recognized by Maple. NOTE: The argument associated with each function is always enclosed in parenthesis ( )'s.

```
> sqrt(x); # square root function
> root[5](x); # 5-th root of x, other order roots similar pattern
> exp(x); # exponential function
> sin(x); # sine function
> cos(x); # cosine function
> tan(x); # tangent function
> sec(x); # secant function
> arcsin(x); # inverse sine function
> arctan(x); # inverse tangent function
> abs(x); # absolute value function
> exp(x); # exponential function
> ln(x); # the natural logarithm function
> log[b](x); # the logarithm function of arbitrary base
```

$$\sqrt{x}$$

$$x^{(1/5)}$$

$$e^x$$

$$\sin(x)$$

$$\cos(x)$$

$$\tan(x)$$

$$\sec(x)$$

$$\arcsin(x)$$

$$\arctan(x)$$

$$|x|$$

$$e^x$$

$$\ln(x)$$

$$\frac{\ln(x)}{\ln(b)}$$

## 1.4 Exercises

1. Calculate the exact and numerical approximations of the the following expressions:

- $\frac{23}{101} + \frac{7}{83} - \frac{47}{211}$ .
- $\sqrt{18} + \sqrt{72}$ .
- The number  $\pi^2$ .

- (d) The number  $e^2$ .
2. The following commands have an error or errors that needs to be corrected. Correct the error so that the proper output is produced.
- (a) Wanting an output of 20, the following is entered:  
`> (2)(10);`
- (b) Recall that  $\cos(0) = 1$ ,  $\sin(\frac{3\pi}{2}) = -1$ , and  $\ln(e) = 1$ . However, the following commands do not produce the desired results.  
`> cos 0; sin(3*pi/2); ln(e^1); ln(exp^1);`
- (c) To factor  $x^2 - 9$ , one enters  
`> FACTOR(x^2-9);`
- (d) To expand  $(x^2 + 1)(x^4 - 1)(x^3 + x^2 + x + 1)$ , one enters  
`> Expand((x^2+1)(x^4-1)(x^3+x^2+x+1));`
- (e) Desiring an answer of  $y$  equal to 7, one enters:  
`> y = 3*x + 1; x := 2; y;`
- (f) For the function  $f(x) = x^3 - x + 1$ , one computes  $f(2)$  by entering:  
`> f := x^3 - x + 1; f(2);`
- (g) For the function  $f(x) = \cos(x^2 - 1)$ , one computes  $f(1)$  by entering  
`> f := x -> cos*(x^2-1); f(1);`

3. Use the **simplify** command to simplify the expression

$$\frac{\frac{1}{x^2} + \frac{x}{y^2}}{\frac{x}{y^2} + \frac{1}{x^2}}.$$

4. Use the **solve** command to solve the following equations:

(a)  $x^3 - 9x^2 + 20x = 0$  for  $x$ .

(b)  $y = \frac{4x - 1}{2x + 1}$  for  $x$ .

(c)  $\cos(2x) = \sin(x)$  for  $x$ .

5. Using the **fsolve** command, solve  $\cos(2 * x) = \sin(x)$  for  $x$  in the interval  $[0, \frac{\pi}{2}]$ .

6. Define  $f(x) = x^3 + x^2 + x + 1$  as a function and compute and simplify  $f(3)$ ,  $f(k^2)$ ,  $f(x - 5)$ , and  $\frac{f(x + h) - f(x)}{h}$ .



## Chapter 2

# More Concerning Functions and Graphing

In this session, we continue to learn more about what we can do with functions defined with Maple. We then introduce Maple's graphing capabilities. To begin, we make sure all variables are unassigned by entering

```
> restart;
```

### 2.1 More about Functions

Recall that we use the  $\rightarrow$  (minus greater than) notation to define a function. Suppose we want to enter the function  $f(x) = x^4 + x^3$ . Suppose, by accident, we forget to enter this as function but instead enter this as an expression.

```
> f := x^4 + x^3;
```

$$f := x^4 + x^3$$

Note, since this is an expression, we cannot find  $f(3)$  properly.

```
> f(3);
```

$$x(3)^4 + x(3)^3$$

We could simply fix this by reentering the expression as a function or use the **unapply** command. The **unapply** command has the form **unapply**(*expression*, *variable*), where *expression* is converted to a function of the specified *variable*. To convert the expression  $f$  stored above to function of  $x$ , we can enter

```
> f := unapply(f, x);
```

$$f := x \rightarrow x^4 + x^3$$

Having  $f$  now expressed as a function now allows us to easily do many different types of function evaluations.

```
> f(3); simplify(f(x-2)); simplify( (f(x+h) - f(x))/h );
```

108

$$x^4 - 7x^3 + 18x^2 - 20x + 8$$
$$4x^3 + 6x^2h + 4xh^2 + h^3 + 3x^2 + 3xh + h^2$$

Recall, to unassign  $f$ , we can enter

```
> f := 'f';
```

$$f := f$$

We can use function notation to define other types of operations we are familiar with, which we now demonstrate.

#### 2.1.1 Even/Odd Function

Recall that a function is odd if  $f(x) = -f(x)$  and even if  $f(-x) = f(x)$ . For example, we can test whether the function  $f(x) = x^4 + x^2$  is even or odd by entering

```
> f := x -> x^4 + x^2;
```

$$f := x \rightarrow x^4 + x^2$$

```
> f(-x); f(x);
```

$$x^4 + x^2$$

$$x^4 + x^2$$

Since  $f(-x) = f(x)$ , the function is even.

Next, we test whether  $g(x) = \sin(x \cos(x))$  is even or odd. First define the function  $g$ .

```
> g := x -> sin(x*cos(x));
```

$$g := x \rightarrow \sin(x \cos(x))$$

We form  $g(-x)$  and  $g(x)$  and compare.

```
> g(-x); g(x);
```

$$\begin{aligned} &-\sin(x \cos(x)) \\ &\sin(x \cos(x)) \end{aligned}$$

Hence,  $g$  is an odd function.

## 2.1.2 Composition

Maple makes it easy to define the composition of functions. Let  $f(x) = \sqrt{x+1}$  and let  $g(x) = x^2 - 4x + 5$ . We enter these functions and form  $f(g(x))$  and  $g(f(x))$ .

```
> f := x -> sqrt(x+1); g := x -> x^2-4*x+5;
```

$$\begin{aligned} f &:= x \rightarrow \sqrt{x+1} \\ g &:= x \rightarrow x^2 - 4x + 5 \end{aligned}$$

Now compute the compositions

```
> f(g(x)); g(f(x));
```

$$\begin{aligned} &\sqrt{x^2 - 4x + 6} \\ &x + 6 - 4\sqrt{x+1} \end{aligned}$$

## 2.1.3 Inverse Function

Maple can be used to calculate the inverse of a function. To demonstrate, suppose we want to find the inverse of the function  $f(x) = \ln(x+1)$ . First, we define the function

```
> f := x -> ln(x+1);
```

$$f := x \rightarrow \ln(x+1)$$

Remember that we need to switch  $x$  and  $y$  and then solve for  $y$ . This means we have to solve the equation  $x = f(y)$  for  $x$ :

```
> solve(x = f(y), y);
```

$$e^x - 1$$

Now define the inverse function.

```
> g := x -> exp(x)-1;
```

$$g := x \rightarrow e^x - 1$$

## 2.2 Graphing

The command that graphs functions in Maple is the **plot** command. The **plot** command has the structure **plot(expression or function, range, options)**, where *expression or function* is the expression or function we want to plot, *range* gives the interval on the  $x$ -axis for which the graph will be plotted, and *options* are specified parameters that are given to add to the graph in a certain way (more details on this will be given below). To do simple plot of the function  $f(x) = x^2 - 4x + 3$ , we can enter

```
> f := x^2 - 4*x + 3;
```

$$f := x^2 - 4x + 3$$

```
> plot(f, x = -2..5);
```

In the preceding command, the function in the first parameter was entered as an expression. To plot as a function, we could also enter

```
> f := x -> x^2-4*x+3;
```




Figure 2.1: Maple graph of  $f(x) = x^2 - 4x + 3$  entered as an expression

$$f := x \rightarrow x^2 - 4x + 3$$

```
> plot(f(x), x = -2..5);
```




Figure 2.2: Maple graph of  $f(x) = x^2 - 4x + 3$  entered as a function

To graph the function in certain ways, we can add more parameters involving options. Here are a few of the options available. These and other options can be seen by specifying `? plot[options]`.

<b>Option</b>	<b>What it Does</b>
<b>view</b> $(x\text{-range}, y\text{-range})$	lets you select the viewing rectangle, like the windows command on a graphing calculator
<b>thickness</b>	the thickness of the curve (values can be 1, 2, or 3 - default is 1), useful if curve is barely visible in a printout.
<b>color</b>	color of the curve. If not specified, a default color is chosen by Maple.
<b>title</b>	title (specified in quotes) of the graph
<b>numpoints</b>	selecting a large number will increase the resolution of the graph and take out artificial corners

If you click on the graph you'll see some additional icons at the top of the screen, which may help to improve the looks of your graph. For the previous function, we can add some of these options as the next command illustrates.

```
> plot(f(x), x = -1..5, view=[-2..5, -2..6], thickness = 3, color = blue, title="my parabola");
```




Figure 2.3: Maple graph of  $f(x) = x^2 - 4x + 3$  with additional parameters

In the previous command, we plotted the function  $f(x) = x^2 - 4x + 3$  with points specified over the interval  $x = -1..5$ . To view the graph, the  $x$  axis ranges from  $-2$  to  $5$  and the  $y$  axis ranges from  $-2$  to  $6$ . The thickness option of  $3$  gives the graph maximum resolution. The color of the graph is specified to be blue with a title of “my parabola”.

There are a couple of methods available for graphing multiple curves on the same graph. One method encloses the functions to be plotted within brackets separated by commas. To demonstrate, suppose we desire to plot the function  $f(x) = \cos(\frac{\pi x}{2})$  in red with a thickness of  $2$ , along with a shift of its graph - shifted up vertically  $2$  units and  $1$  unit to the right horizontally, in blue with a thickness of  $1$ . We plot these functions on the interval  $[-\pi, 3\pi]$ . The following command accomplishes this task.

```
> f := x -> cos(Pi*x/2);
      f := x -> cos( $\frac{1}{2} \pi x$ )
> plot( [f(x), f(x-1)+2], x = -Pi..3*Pi, color=[red,blue], thickness=[2,1]);
```




Figure 2.4: Graph of cosine function  $\cos(\frac{\pi}{2}x)$  with horizontal shift of one a vertical shift of two

In preceding command, you should notice that the functions specified in brackets, their colors, and their thicknesses, are specified correspondedly in the same order.

The next example illustrates how a function and its inverse are displayed on common graphs. We first enter the function we want to find the inverse of:

```
> f := x -> ln(x+1);
```

$$f := x \rightarrow \ln(x + 1)$$

Using the **solve** command, we can find the inverse function:

```
> g := solve( x=f(y),y);
```

$$g := e^x - 1$$

The previous command outputs an expression. To convert this to a function of x, we can use the **unapply** command

```
> g := unapply(g, x);
```

$$g := x \rightarrow e^x - 1$$

The next command graphs both the function and inverse on the same graph.

```
> plot( [f(x),g(x)], x = -2.5..2.5, view=[-2.5..2.5, -2.5..2.5], color = [red, green],
> thickness = [1,2], title = "Graph of f(x) = ln(x+1)and its inverse g(x) = e^x - 1");
```




Figure 2.5: Maple graph of  $f(x) = \ln(x + 1)$  and inverse  $g(x) = e^x - 1$

Another method for plotting the graph involves storing each separate graph plots as variables and displaying them together. This feature especially convenient when one wants to graph functions over different intervals on the x axis. To demonstrate, suppose we enter the following two functions:

```
> f := x -> x^2; g := x -> x^5;
```

$$f := x \rightarrow x^2$$

$$g := x \rightarrow x^5$$

The following commands set up and store the plots for these graphs in the variables p1 and p2.

**NOTE!!** An important fact to remember is to end these commands with a colon : . If you end the command with a semicolon, all the data points generated to plot the graph will be displayed.

```
> p1 := plot(f(x), x = -3..3, color = blue):
```

```
> p2 := plot(g(x), x = -1.2..1.2, color = black, thickness = 2):
```

Note the different interval values for x that are specified. To plot the two graphs, we must specify a command in the Maple **plots** package. The **plots** package is not available as a standard part of Maple and must be read in. This can be done with the following statement

```
> with(plots);
```

Warning, the name changecoords has been redefined

[*animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, cylinderplot, densityplot, display, display3d, fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra\_supported, polyhedraplot, replot, rootlocus, semilogplot, setoptions, setoptions3d, spacecurve, sparsematrixplot, sphereplot, surfdata, textplot, textplot3d, tubeplot*]

The output of the preceding statement gives all the new commands that we now have available to use. The one we need is the **display** command. To display the plots stored in `p1` and `p2`, we enter

```
> display([p1, p2], view = [-4..4, -5..10], title = "f(x) = x^2 and g(x) = x^5 with
> different intervals");
```




Figure 2.6: Maple graph of  $f(x) = x^2$  and  $g(x) = x^5$  with different intervals

Note in the preceding **display** command that the same options, such as `view` and `title`, can be specified like we did in the **plot** command

## 2.3 Exercises

- Using the **unapply** command, convert the assignment to a function and compute  $f(\frac{3\pi}{4})$ .

```
> f := tan(x) + sec(x);
```

- Determine whether the following functions are even, odd, or neither.

(a)  $f(x) = x^3 + x$ .

(b)  $f(x) = \sin(\cos(x)) + \sec(x)$ .

(c)  $f(x) = x^3 + e^x$ .

- Determine  $f(g(x))$  and  $g(f(x))$  for the following functions.

(a)  $f(x) = x^3 + x^2 + 1$  and  $g(x) = 2 * x^4 + x$ .

(b)  $f(x) = \arcsin(x^2)$  and  $g(x) = \sqrt{\sin(x)}$ .

- Find the inverse of the function  $f(x) = e^{x-1}$ .

- Using the **plot** command, graph the following functions:

(a)  $f(x) = x^2 - 2x - 3$ .

(b)  $g(x) = e^{\sin(x)}$ .

6. Using the **plot** command, plot the following functions on the same graph. Color one function red with a thickness of 2 and the other blue with a thickness of 1. Create a title for your graph.
- (a)  $f(x) = x^2$  and  $g(x) = x^3$ .
  - (b)  $f(x) = \ln(x^2)$  and  $g(x) = e^{\sin(x)}$ .
  - (c)  $f(x) = \sin(\frac{3\pi}{2})$  and the graph resulting from shifting the graph up 3 units and to the right 2 units.
  - (d) The graph of  $f(x) = e^{x-1}$  and its inverse.
7. By storing your plots and using the display command, plot the graph of  $f(x) = e^{x^2}$  on the interval  $-2 \leq x \leq 2$  and the graph of  $g(x) = \sin(x^2)$  on the interval  $-2\pi \leq x \leq 2\pi$ .

